X-721-66-162

NASA TM X-55523

FIN
A COMPUTER PROGRAM FOR CALCULATING THE AERODYNAMIC CHARACTERISTICS OF FINS AT SUPERSONIC SPEEDS

APRIL 1966

NASA

GODDARD SPACE FLIGHT CENTER
GREENBELT, MARYLAND

X-721-66-162

FIN

A COMPUTER PROGRAM FOR CALCULATING THE AERODYNAMIC CHARACTERISTICS OF FINS AT SUPERSONIC SPEEDS

James S. Barrowman
Spacecraft Integration and Sounding Rocket Division

April 1966

GODDARD SPACE FLIGHT CENTER
Greenbelt, Maryland

FIN

A COMPUTER PROGRAM FOR CALCULATING THE AERODYNAMIC CHARACTERISTICS OF FINS AT SUPERSONIC SPEEDS

James. S. Barrowman
Spacecraft Integration and Sounding Rocket Division

30346

SUMMARY

By numerical solution of Busemann's Second Order Airfoil Theory and the spanwise summing of airfoil strips, FIN determines the pressure coefficient distributed over a given fin configuration moving at supersonic speeds. In determining the distribution, the program can include the effect of a fin-tip Mach cone.

From this basic calculation, FIN can determine as functions of angle of attack the lift coefficient, wave drag coefficient, pitching moment coefficient, and center-of-pressure location, and as a function of fin cant angle the rolling moment coefficient. FIN can also determine the lift coefficient slope, wave drag at zero angle of attack, pitching moment coefficient slope, rolling moment coefficient slope, and center-of-pressure location at zero angle of attack as functions of Mach number.

Comparisons with windtunnel data show that predicted values using FIN output fall well within 10% of experiment results.

CONTENTS

ILLUSTRATIONS

LIST OF SYMBOLS

A	Reference area
C	Length of airfoil strip chord
C_{DW}	Wave drag coefficient $= \dfrac{F_d}{qA}$
C_L	Lift coefficient $= \dfrac{F_l}{qA}$
C_{L_α}	$dC_L/d\alpha(\alpha = 0)$
C_l	Rolling moment coefficient $= \dfrac{Mr}{qAL}$
C_{l_δ}	$dC_l/d\delta(\delta = 0)$
C_m	Pitching moment coefficient $= \dfrac{Mp}{qAL}$
C_{m_α}	$dC_m/d\alpha(\alpha = 0)$
C_{P_i}	Local pressure coefficient
C_r	Length of root chord
d_i	Component of l_i parallel to the freestream
d_w	Portion of d_i behind the fin-tip Mach cone
F_d	Drag force (parallel to freestream)
F_l	Lift force (normal to freestream)
K	Interference factor
L	Reference length
l_L	Length of airfoil-strip leading-edge region chord
l_{L_r}	Length of root airfoil leading-edge region chord
l_M	Length of center region and of center-region chord
l_T	Length of airfoil-strip trailing-edge region chord
l_{T_r}	Length of root airfoil trailing-edge region chord

LIST OF SYMBOLS (Cont.)

l_w	Distance between the leading edge and the intersection of the Mach cone with the airfoil strip
l_i	Length of local region surface
M	Freestream Mach number
M_P	Pitching moment about the reference axis
M_r	Rolling moment about the root chord
n	Number of strips
n_i	Component of l_i normal to the freestream
P_i	Local static pressure
P_∞	Ambient static pressure
q_∞	Freestream dynamic pressure
r_i	d_w / d_i ratio
S	Semispan length
X_L	Distance from the reference axis to the airfoil-strip leading edge in a streamwise direction
$\bar{X}$	Center-of-pressure coordinate measured from the reference axis
x_{Pi}	Distance from the reference axis to the forward region boundary of a local region in a streamwise direction
$\bar{Y}$	Center-of-pressure coordinate measured from the root chord
y	Distance from the root chord in a spanwise direction
Δy	Width of airfoil strip
α	Angle of attack (degree)

LIST OF SYMBOLS (Cont.)

β	$\sqrt{M^2 - 1}$
Γ_L	Leading-edge sweep angle
Γ_T	Trailing-edge sweep angle
Γ_1, Γ_2	Region boundaries sweep angles
γ	Ratio of specific heats for air = 1.4
δ	Fin cant angle
ζ_L	Leading-edge wedge half-angle in the streamwise direction
ζ_T	Trailing-edge wedge half-angle in the streamwise direction
η	Inclination of a local surface to the freestream
μ	Mach cone semivertex angle

SUBSCRIPTS

B	Body alone
B(T)	Body in the presence of the tail
i	Local region number
j	Airfoil strip number
o	Value at $\alpha = 0$
T	Tail alone
T(B)	Tail in the presence of the body

BLANK PAGE

FIN

A COMPUTER PROGRAM FOR CALCULATING THE AERODYNAMIC CHARACTERISTICS OF FINS AT SUPERSONIC SPEEDS

BACKGROUND

Calculating the aerodynamic characteristics of the fins of a sounding rocket is one of the basic steps in calculating the overall aerodynamics of a sounding rocket. Because sounding rockets fly at supersonic speeds for the greater portion of their flight, the regime at these speeds is of particular interest.

Methods previously used for calculating supersonic fin aerodynamics (references 1, 2, 3, and 4) are lacking in accuracy and applicability because of the many assumptions and approximations inherent in using them to reach a closed form or nearly closed form solution.

By using a high-speed computer to numerically solve basic theoretical equations, one may obtain answers rapidly and as accurately as desired. The only restrictions to accuracy are the assumptions inherent in deriving the basic equations. The theory found to be most amenable to programming is Busemann's Second Order Supersonic Airfoil Theory as described in reference 1.

PROGRAM CAPABILITIES AND LIMITATIONS

Given a supersonic fin with chord plane symmetry, at a given Mach number, FIN computes as functions of angle of attack (α) the following:

- Lift coefficient, C_L
- Wave drag coefficient, C_{DW}
- Pitching moment coefficient, C_m
- Center-of-pressure coordinate measured from the reference axis, $\bar{X}$
- Center-of-pressure coordinate measured from the root chord, $\bar{Y}$

Since α as defined in the program is equivalent to the fin cant angle (δ) at $\alpha = 0$, the rolling moment coefficient C_l is considered a function of δ.

FIN also computes as functions of Mach number the following:

- Lift coefficient slope, C_{L_α}
- Wave drag at zero angle of attack, C_{DW_o}
- Pitching moment coefficient slope, C_{m_α}
- Rolling moment coefficient slope, C_{l_δ}
- Center-of-pressure coordinate measured from the reference axis at zero angle of attack, $\bar{X}_0$
- Center-of-pressure coordinate measured from the root chord at zero angle of attack, $\bar{Y}_0$

The range of α is from zero to α_{max}, as determined by the user, in increments of 1 degree. The program allows for a maximum of 100 Mach number points, the user specifying the range and increment of the Mach number.

Since FIN has been designed for use with a four-finned vehicle, the output values of C_L, C_{DW}, C_m, C_{L_α}, C_{DW_o}, and C_{m_α} are for a pair of identical fins; the output values of C_l and C_{l_δ} are for two pairs of perpendicular fins. The values of $\bar{X}$, $\bar{Y}$, $\bar{X}_o$, and $\bar{Y}_o$ apply only to a single typical fin.

Busemann's Second Order Airfoil Theory is applied subject to the restrictions outlined in reference 1, pages 192 to 241, with special attention to Figures 10.3 and 10.28. However, the use of the third-order terms in this program slightly enlarges the applicability shown in this reference. Busemann's theory is applied to small streamwise airfoil strips, and the strips are summed in a spanwise direction. In addition, the fin-tip Mach cone is accounted for by applying a correction factor to the necessary portion of each strip. This correction-factor technique was obtained from references 1 and 3. The fin-tip Mach cone correction may or may not be included at the user's discretion.

Figure 1 shows the types of fins to which FIN is applicable. These configurations cover all of the shapes normally used on sounding rockets and missiles. The program input pattern for each type is given in the section on usage. A listing of FIN in FORTRAN IV is given in Appendix A.

AERODYNAMIC THEORY

Busemann's Second Order Airfoil Theory has been applied to two-dimensional airfoil strips with the following assumptions:



(a) Modified Double Wedge

(b) Double Wedge

(c) Modified Single Wedge

(d) Single Wedge

Figure 1. Fin Configurations

- All parts of the airfoil surface are in supersonic flow and make small angles with the flow. This implies low angles of attack.

- The leading edge is sharp. The trailing edge must be sharp for a double wedge fin or a modified double wedge. For a single wedge and a modified single wedge, the effect of the blunt base is neglected.

- The shock waves are all attached.

- Each region of flow over the surface acts independently of the others.

The basic local pressure coefficient equation of the Busemann theory is a third-order series expansion:

$$C_{P_i} = \frac{P_i - P_\infty}{q_\infty} = K_1 \eta_i + K_2 \eta_i^2 + K_3 \eta_i^3 - K^* \eta_L^3 \tag{1}$$

where

$$K_1 = \frac{2}{\beta} \tag{2a}$$

$$K_2 = \frac{(\gamma + 1)M^4 - 4\beta^2}{4\beta^4} \tag{2b}$$

$$K_3 = \frac{(\gamma + 1)M^8 + \left(2\gamma^2 - 7\gamma - 5\right)M^6 + 10(\gamma + 1)M^4 - 12M^2 + 8}{6\beta^7} \tag{2c}$$

If the flow over the surface region inclined at η_i to the flow has been preceded by a single compressive shock wave anywhere upstream in the flow, the value of K^* is:

$$K^* = \frac{(\gamma + 1)M^4 \left[(5 - 3\gamma)M^4 + 4(\gamma - 3)M^2 + 8\right]}{48} \tag{3}$$

If the flow upstream of the region in question has contained only expansive shock-free flow, $K^* = 0$. η_L is the inclination of the leading edge surface. Essentially, the first three terms of equation 1 represent the first-, second-, and third-order pressures on the surface region respectively. The K^* term is the third-order irreversible pressure rise through the bow shock. The index, i, indicates the surface for which the pressure coefficient is being calculated.

The pressure coefficient of equation 1 is applied to a typical airfoil region, including the fin-tip Mach cone, by dividing the region into two portions at the intersection of the airfoil strip and the Mach cone (Figure 2). A correction factor of ½ is applied to the local pressure coefficient for the region within the Mach cone. The center of pressure of each portion is taken as the midpoint of the length of that portion. By this means, equations are obtained for the local lift $\left(F_{l_i}\right)$, drag $\left(F_{d_i}\right)$, local center of pressure $\left(\bar{X}_i\right)$, and, hence, the local pitching moment $\left(M_{p_i}\right)$:

$$F_{l_i} = C_{P_i} \, d_i \left(1 - \frac{r_i}{2}\right) \tag{4}$$

$$F_{d_i} = C_{P_i} \, n_i \left(1 - \frac{r_i}{2}\right) \tag{5}$$

$$\bar{X}_i = \frac{\tfrac{1}{2} d_i \left[1 - r_i + \dfrac{r_i}{2} + \dfrac{x_{p_i}}{l_i}\left(2 - r_i\right)\right]}{1 - \dfrac{r_i}{2}} \tag{6}$$

$$M_{p_i} = F_{l_i} \, \bar{X}_i \tag{7}$$

The airfoil-strip lift $\left(F_{l_j}\right)$, drag $\left(F_{d_j}\right)$, and pitching moment $\left(M_{P_j}\right)$ totals are then computed by summing the local characteristics.

$$F_{l_j} = \sum_{i=1}^{6} F_{l_i} \tag{8}$$

$$F_{d_j} = \sum_{i=1}^{6} F_{d_i} \tag{9}$$

$$M_{P_j} = \sum_{i=1}^{6} M_{P_i} \tag{10}$$



Figure 2. Mach Cone Correction Geometry

The strip rolling moment about the root chord is calculated from the airfoil lift and strip spanwise position, y :

$$M_{r_j} = yF_{l_j} \tag{11}$$

The total characteristics over the entire fin are then computed by summing the airfoil-strip characteristics in a spanwise direction.

$$F_l = \sum_{j=1}^{n} F_{l_j} \tag{12}$$

$$F_d = \sum_{j=1}^{n} F_{d_j} \tag{13}$$

$$M_p = \sum_{j=1}^{n} M_{p_j} \tag{14}$$

$$M_r = \sum_{j=1}^{n} M_{r_j} \tag{15}$$

The forces and moments computed in equations 4 through 15 are actually divided by the ambient dynamic pressure (q_∞) and strip width (Δy) . Thus, the associated coefficients are:

$$C_L = \frac{2F_l \cdot \Delta y}{A} \qquad \text{(2 fins)} \tag{16}$$

$$C_{DW} = \frac{2F_d \cdot \Delta y}{A} \qquad \text{(2 fins)} \tag{17}$$

$$C_m = \frac{2M_p \cdot \Delta y}{AL} \qquad \text{(2 fins)} \tag{18}$$

$$C_l = \frac{4M_r \cdot \Delta y}{AL} \qquad \text{(4 fins)} \tag{19}$$

The center-of-pressure coordinates are calculated from the respective moments and the appropriate lift forces.

$$\bar{X} = \frac{M_p}{F_1 \cos \alpha} \qquad (20)$$

$$\bar{Y} = \frac{M_r}{F_1} \qquad (21)$$

To obtain the linear slope coefficients near $\alpha = 0$ at a given Mach number, the values of C_L, C_m, and C_l are calculated at $\alpha = 1°$ and are taken as the slope values $C_{L\alpha}$, $C_{m\alpha}$, and $C_{l\delta}$ (all per degree). The slope values per radian are obtained by the proper conversion of radians to degrees. Thus:

$$\frac{C_{L_\alpha}}{\text{rad.}} = \frac{180}{\pi}\left(C_L\big|_{\alpha = 1}\right) \qquad (22)$$

$$\frac{C_{m_\alpha}}{\text{rad.}} = \frac{180}{\pi}\left(C_m\big|_{\alpha = 1}\right) \qquad (23)$$

$$\frac{C_{l_\delta}}{\text{rad.}} = \frac{180}{\pi}\left(C_l\big|_{\delta = 1}\right) \qquad (24)$$

The zero angle-of-attack values of C_{DW}, $\bar{X}$, and $\bar{Y}$ are simply their respective values at $\alpha = 0$.

CONFIGURATION AND GEOMETRY CONSIDERATIONS

The user indicates the shape of the fin configuration by specifying the root chord length (C_r), root airfoil leading and trailing edge region chord lengths $(l_{Lr}$ and $l_{Tr})$, leading and trailing edge sweep angles $(\Gamma_L$ and $\Gamma_T)$, region boundary sweep angles $(\Gamma_1$ and $\Gamma_2)$, and the fin semispan length (S) (Figure 3). The program then computes the distance from the reference axis to the airfoil-strip leading edge in the streamwise direction (X_L), airfoil chord length (C), and the airfoil region chord lengths (l_L and l_T) as functions of the spanwise coordinate of the airfoil strip.

$$X_L = y \tan\Gamma_L \cos \alpha \qquad (25)$$

$$C = y\left(\tan\Gamma_T - \tan\Gamma_L\right) + C_r \qquad (26)$$



Figure 3. Fin and Airfoil Geometry

$$l_L = y\left(\tan\Gamma_1 - \tan\Gamma_L\right) + l_{L_r} \tag{27}$$

$$l_T = y\left(\tan\Gamma_T - \tan\Gamma_2\right) + l_{T_r} \tag{28}$$

The width of the airfoil strip is determined from the number of strips (n) desired by the user.

$$\Delta y = \frac{S}{n} \tag{29}$$

The airfoil strip crossection is assumed to be symmetrical about the chord line. The most general crossection configuration is the modified double wedge (Figure 1a). The other three types (Figures 1b, 1c, and 1d) are special cases of this basic shape. Figure 4 shows in detail the general crossection, which has six separate flow regions. As can be seen in this figure, the surface region lengths are determined by specifying the leading- and trailing-edge wedge half angles in the streamwise direction, ζ_L and ζ_T.

$$l_1 = l_4 = \frac{l_L}{\cos\zeta_L} \tag{30}$$

$$l_3 = l_6 = \frac{i_T}{\cos\zeta_T} \tag{31}$$

$$l_2 = l_5 = C - l_L - l_T \tag{32}$$

The program computes the local-surface inclination angles $\left(\eta_i\right)$ as functions of α:

$$\eta_1 = \zeta_L - \alpha \tag{33a}$$

$$\eta_2 = -\alpha \tag{33b}$$

$$\eta_3 = -\zeta_T - \alpha \tag{33c}$$

$$\eta_4 = \zeta_L + \alpha \tag{33d}$$



Figure 4. Airfoil Angles

$$\eta_5 = \alpha \tag{33e}$$

$$\eta_6 = -\zeta_T + \alpha \tag{33f}$$

From these, the streamwise and normal components of the region lengths are calculated.

$$\begin{aligned} d_i &= l_i \cos \eta_i \quad (i = 1 \rightarrow 6) \\ n_i &= l_i \sin \eta_i \quad (i = 1 \rightarrow 6) \end{aligned} \tag{34}$$

The distance of the forward local-region boundaries from the reference axis are then determined as:

$$x_{P1} = x_{P4} = X_L \tag{35a}$$

$$x_{P2} = X_L + d_1 \tag{35b}$$

$$x_{P3} = X_L + d_1 + d_2 \tag{35c}$$

$$x_{P5} = X_L + d_4 \tag{35d}$$

$$x_{P6} = X_L + d_4 + d_5 \tag{35e}$$

FIN-TIP MACH CONE CORRECTION

The Mach-cone angle (μ) is a direct function of the Mach number.

$$\mu = \text{Arctan}\left(\frac{1}{\beta}\right) = \text{Arcsin}\left(\frac{1}{M}\right) \tag{36}$$

For an airfoil strip at a spanwise distance y, the intersection of the strip and the Mach cone is a distance l_w from the leading edge.

$$l_w = (S - y)\tan\Gamma_L + \tan(90 - \mu) \tag{37}$$

But, by trigonometric identities and the use of equation 37, this may be reduced to:

$$l_W = (S - y)\left(\tan \Gamma_L + \beta\right) \tag{38}$$

If l_W is greater than or equal to the chord length, there is, of course, no actual intersection and no correction is necessary. If l_W is such that

$$C > l_W \geq C - l_T$$

then the ratio of the portion of l_T falling behind the Mach cone to l_T is:

$$r_3 = r_6 = \frac{C - l_W}{l_T} \tag{39}$$

Also,

$$r_1 = r_2 = r_4 = r_5 = 0 \tag{40}$$

If l_W is such that

$$C - l_T > l_W \geq l_L$$

then the corresponding ratio for the middle regions is

$$r_2 = r_5 = \frac{C - l_T - l_W}{l_M} \tag{41}$$

Also,

$$r_3 - r_6 = 1 \tag{42a}$$

and,

$$r_1 = r_4 = 0 \tag{42b}$$

If l_W is such that

$$C - l_T - l_M > l_W \geq 0$$

then the ratio for the leading edge regions is

$$r_1 = r_4 = \frac{l_L - l_W}{l_L} \tag{43}$$

Also,

$$r_2 = r_3 = r_5 = r_6 = 1.0 \tag{44}$$

These ratios are the values used in equations 4, 5, and 6 for calculating the local lift, drag, and center of pressure. In these equations, the pressure coefficient in the portion behind the Mach cone is taken as half the value of the C_p calculated by equation 1.

PROGRAM USAGE

The input data required by FIN are:

M_o	Initial Mach number
ΔM	Mach number increment
M_{max}	Maximum Mach number
α_{max}	Maximum angle of attack
A	Reference area
L	Reference length
C_r	Root chord length
S	Fin semispan
l_{L_r}	Root airfoil leading-edge region chord length
l_{T_r}	Root airfoil trailing-edge region chord length
Λ_L	Leading-edge sweep angle
Λ_1, Λ_2	Region boundary sweep angles
Λ_T	sweep angle
	in the streamwise direction
	direction

n	Number of strips desired
k = 0	Compute with fin tip Mach cone correction
= 1	Compute without Mach cone correction

Lengths and areas must be input in any consistant units. Angles are all input in degrees.

Four input cards are required for each run. The input data are arranged in the order shown on the indicated card and the cards must be in the indicated order:

- Card 1: Any 72 Hollerith characters for run identification
- Card 2: M_o, ΔM, M_{max}, α_{max}, A, L, C_r, S
- Card 3: l_{L_r}, l_{T_r}, Γ_L, Γ_1, Γ_2, Γ_T, ζ_L, ζ_T
- Card 4: n, k

Any number of four-card sets may be input.

Cards 2 and 3 contain eight fields each having ten spaces. Variables may be placed anywhere in their respective field, but each value on cards 2 and 3 contain a decimal point. Card 4 contains two fields each having three spaces. The values on card 4 do not contain a decimal point; however, the number must always end in the third space in the field.

Appendix B shows a typical set of inputs.

The user may exercise a number of calculation options by proper definition of certain input parameters. If $M_o = 0$, then the program sets:

$$M_o = \mu_o^2 + 1 \tag{45}$$

where

$$\mu_o = \frac{C_r}{S} - \tan \Gamma_L \tag{46}$$

However, if $\mu_o \leq 0$, the program sets $M_o = 1$. This option provides the minimum allowable Mach number for the given configuration. If $\alpha_{max} = 1$, the program outputs only the values of C_{L_α}, C_{DW_o}, C_{m_α}, C_{l_δ}, $\overline{X}$, and $\overline{Y}$ as functions of Mach number. As indicated in the input list, the fin-tip Mach cone may or may not be included by setting k equal to zero or one respectively.

For a modified double wedge fin, the input values are simply those for the particular configuration. For the special cases of the modified double wedge, however, the configuration data must be input according to the following relations:

Double Wedge

Γ_L = input value	C_r = input value
Γ_1 = input value	l_{L_r} = input value
$\Gamma_2 = \Gamma_1$	$l_{T_r} = C_r - l_{L_r}$
Γ_T = input value	ζ_L = input value
	ζ_T = input value

Modified Single Wedge

Γ_L = input value	C_r = input value
Γ_1 = input value	l_{L_r} = input value
$\Gamma_2 = \Gamma_T$	$l_{T_r} = 0$
Γ_T = input value	ζ_L = input value
	$\zeta_T = 0$

Single Wedge

Γ_L = input value	C_r = input value
$\Gamma_1 = \Gamma_T$	$l_{L_r} = C_r$
$\Gamma_2 = \Gamma_T$	$l_{T_r} = 0$
Γ_T = input value	ζ_L = input value
	$\zeta_T = 0$

RESULTS AND CONCLUSIONS

To date, FIN has been applied to the fins of the Aerobee 150A, Aerobee 350, and Tomahawk sounding rockets. Of these, only the first two have enough reliable windtunnel data to allow meaningful comparison (references 5 and 6). Unfortunately, the data for these vehicles did not include enough parameters to allow a comparison of all the calculated characteristics. However, since all the characteristics stem from the same basic C_p distribution, it is reasonable to infer the degree of accuracy, to a first approximation, of all the calculated quantities from the comparison of a few key characteristics. The characteristics that were compared to windtunnel data are C_L, C_{L_α}, and $\overline{X}_o$. The comparison calculations are not included in FIN.

Both the Aerobee 150A and Aerobee 350 data contained total vehicle and body-alone data for C_L, C_{L_α}, and $\overline{X}_o$. By combining the windtunnel body-alone data with FIN tail data, total vehicle values may be calculated for comparison with the total vehicle windtunnel data. The total vehicle parameters are obtained from the contributions of separate portions of the vehicle.

$$C_L = \left(C_L\right)_B + \left(C_L\right)_{B(T)} + \left(C_L\right)_{T(B)}$$

$$C_{L_\alpha} = \left(C_{L_\alpha}\right)_B + \left(C_{L_\alpha}\right)_{B(T)} + \left(C_{L_\alpha}\right)_{T(B)}$$

$$\overline{X}_o = \frac{\left(C_{L_\alpha}\right)_P\left(\overline{X}_o\right)_B + \left(C_{L_\alpha}\right)_{B(T)}\left(\overline{X}_o\right)_{B(T)} + \left(C_{L_\alpha}\right)_{T(B)}\left(\overline{X}_o\right)_{T(B)}}{C_{L_\alpha}}$$

The contributions of the body in the presence of the tail, $\left(C_L\right)_{B(T)}$, $\left(C_{L_\alpha}\right)_{B(T)}$ and $\left(\overline{X}_o\right)_{B(T)}$, are determined by the method of reference 2, using charts 4b and 14b. The contributions of the tail in the presence of the body, $\left(C_L\right)_{T(B)}$, $\left(C_{L_\alpha}\right)_{T(B)}$, and $\left(\overline{X}_o\right)_{T(B)}$, are calculated from the FIN values $\left(C_L\right)_T$, $\left(C_{L_\alpha}\right)_T$, and $\left(\overline{X}_o\right)_T$ by applying the tail-body interference factor technique from reference 2.

$$\left(C_L\right)_{T(B)} = K_{T(B)}\left(C_L\right)_T \tag{47}$$

$$\left(C_{L_\alpha}\right)_{T(B)} = K_{T(B)}\left(C_{L_\alpha}\right)_T$$

$$\left(\overline{X}_o\right)_{T(B)} = \left(\overline{X}_o\right)_T$$

where $K_{T(B)}$ is the interference factor given by equation 14 and chart 1 of reference 2. Since this factor is valid only near $\alpha = 0$, equation 47 is valid only for small angles of attack. Figure 5 shows C_L vs. α for the Aerobee 150A at Mach 3.01. Figures 6 and 7 show C_{L_α} vs. M for the Aerobee 150A and Aerobee 350 respectively. Figures 8 and 9 show the respective curves of $\overline{X}_o$ vs. M.

As the figures show, the values calculated using FIN output are well within 10% of all the windtunnel values at low angles of attack between Mach 2 and 7. This represents adequate prediction for use in first-look trajectory analyses or for determining requirements for windtunnel tests on a new or proposed sounding rocket vehicle.



Figure 5. Lift Coefficient as a Function of Angle of Attack, Aerobee 150A at Mach 3.01



Figure 6. Lift Coefficient Slope as a Function of Mach Number, Aerobee 150A



Figure 7. Lift Coefficient Slope as a Function of Mach Number, Aerobee 350



Figure 8. Center of Pressure from Nose as a Function of Mach Number, Aerobee 150A



Figure 9. Center of Pressure from Nose as a Function of Mach Number, Aerobee 350

APPENDIX A

PROGRAM LISTING
OF FIN IN
FORTRAN IV

PRECEDING PAGE BLANK NOT FILMED.

```
$JOB   1091P003 405J SNOW BARROWMAN
$EXECUTE          IBJOB
$IBJOB            GO,SOURCE,NOMAP
$IBFTC FIN        M94,XR7,DECK
C FIN
C CALCULATION OF,
C        CL X ALPHA
C        CM X ALPHA
C        CL X DELTA (ROLLING MOMENT COEFF.)
C        CDW X ALPHA
C        CENTER OF PRESSURE X ALPHA
C        LIFT CURVE SLOPE
C        PITCHING MOMENT CURVE SLOPE
C        ROLLING MOMENT CURVE SLOPE
C        WAVE DRAG CURVE SLOPE
C   OF A THREE DIMENSIONAL FIN.  PROGRAM USES BUSEMANNS SECOND ORDER
C   THEORY WITH THE THIRD ORDER TERMS RETAINED.  TWO DIMENSIONAL VALUES
C   ARE CALCULATED ALONG STREAMWISE STRIPS AND ARE THEN SUMMED IN A
C   SPANWISE DIRECTION.  THE EFFECT OF THE FIN TIP MACH CONE IS
C   CONSIDERED BY DECREASING THE LOCAL PRESSURE COEFFICIENT BY A
C   FACTOR OF 1/2 WHEN THE POINT IS WITHIN THE MACH CONE.
C    GOOD FOR WEDGE, DOUBLE WEDGE, OR MODIFIED DOUBLE WEDGE AIRFOILS
C   WITH SYMMETRY ABOUT THE CHORD LINE.
      DIMENSION ETA(6),ETAR(6),CDO(100),CLA(100),TLAM(4),LAM(4),N(6)
      DIMENSION XF(6),CLD(100),CMA(100),CPXX(100),CPSS(100)
      DIMENSION PL(6),D(6),R(6),FL(6),FD(6),TITLE(12),M(100)
      REAL M,N,LAM,LW,MO,K1,K2,K3,LL,LT,LIFT,MMAX,MSQ,LAR,MU
      REAL LLR,LTR,MO,MMOMT,LMOM
      DTR=3.1415927/180.0
1     READ (5,2) (TITLE(I),I=1,12)
2     FORMAT (12A6)
      READ (5,3) MO,DELM,MMAX,AMAX,AREA,REFL,CR,SPAN
      READ (5,3) LLR,LTR,(LAM(I),I=1,4),ZETAL,ZETAT
3     FORMAT (8E10.4)
      READ (5,4) NS,I2
4     FORMAT (2I3)
      WRITE (6,5)
5     FORMAT (1H1)
      WRITE (6,2) (TITLE(I),I=1,12)
C   FIN ANGLE VALUES
      DO 6 I=1,4
      LAR=LAM(I)*DTR
6     TLAM(I)=SIN(LAR)/COS(LAR)
      ZETALR=ZETAL*DTR
      ZETATR=ZETAT*DTR
      CZL=COS(ZETALR)
      CZT=COS(ZETATR)
      WRITE (6,7) ZETAL,ZETAT
7     FORMAT (11HO ZETA L = F6.3,5H DEG.,5X,9HZETA T = F6.3,5H DEG.)
      WRITE (6,8) AREA,REFL,NS
8     FORMAT (13HO      AREF = F8.2,9H  LREF = F8.3,11H  STRIPS = I3)
C   STRIP WIDTH
      XNS=NS
      DS=SPAN/XNS
      XNS=2.*DS/AREA
C   INITIAL MACH NUMBER
      J=1
      IF (MO.NE.0.) GOTO10
      CMU=CR/SPAN-TLAM(1)
      IF (CMU.LE.0.) GOTO9
      M(J)=SQRT(CMU*CMU+1.)
      GO TO 12
9     M(J)=1.0
```

PRECEDING PAGE BLANK NOT FILMED

```
      GO TO 12
10    M(J)=MO
      GO TO 12
C   MACH NUMBER
11    J=J+1
      M(J)=M(J-1)+DELM
      IF (M(J).GT.MMAX) GOTO43
12    MSQ=M(J)*M(J)
      BETA=SQRT(MSQ-1.)
      DEM=-2.*MSQ+4./3.
      DEN=1./BETA**7
      K1=2./BETA
      K2=(1.2*MSQ**2-2.*BETA**2)/BETA**4
      K3=(.4*MSQ**4-(10.88/6.)*MSQ**3+4.*MSQ*MSQ+DEM)*DEN
      B3=(.04*MSQ**4-.32*MSQ**3+.4*MSQ*MSQ)*DEN
C   MACH ANGLE
      CMU=(TLAM(1)+BETA)
      MU=SPAN*CMU
      IF (AMAX.EQ.1.) GOTO16
      WRITE (6,13)
13    FORMAT (1HO)
      WRITE (6,14) M(J)
14    FORMAT (5HOM = F6.3)
      WRITE (6,15)
15    FORMAT (58H    ALPHA      CL        CM       CLM       CDW       CPX
     1CPS)
C CL VS. ALPHA CURVE
16    NAMAX=AMAX+1.
      DO 42 K=1,NAMAX
      ALPHA=K-1
      ALPHAR=ALPHA*DTR
      TLIFT=0.
      TDRAG=0.
      TMMOM=0.
      TLMOM=0.
      S=-DS
17    S=S+DS
      C=S*(TLAM(4)-TLAM(1))+CR
      LL=S*(TLAM(2)-TLAM(1))+LLR
      LT=S*(TLAM(4)-TLAM(3))+LTR
      SL=LL/CZL
      ST=LT/CZT
      SM=C-LL-LT
      XL=S*TLAM(1)*COS(ALPHAR)
      ETA(1)=ZETAL-ALPHA
      ETA(2)=-ALPHA
      ETA(3)=-ZETAT-ALPHA
      ETA(4)=ZETAL+ALPHA
      ETA(5)=ALPHA
      ETA(6)=-ZETAT+ALPHA
      DO 18 I=1,6
18    ETAR(I)=ETA(I)*DTR
      D(1)=SL*COS(ETAR(1))
      D(2)=SM*COS(ETAR(2))
      D(3)=ST*COS(ETAR(3))
      D(4)=SL*COS(ETAR(4))
      D(5)=SM*COS(ETAR(5))
      D(6)=ST*COS(ETAR(6))
      N(1)=SL*SIN(ETAR(1))
      N(2)=SM*SIN(ETAR(2))
      N(3)=ST*SIN(ETAR(3))
      N(4)=SL*SIN(ETAR(4))
      N(5)=SM*SIN(ETAR(5))
      N(6)=ST*SIN(ETAR(6))
      DO 21 I=1,6
      SAVE=K1*ETAR(I)+K2*ETAR(I)**2+K3*ETAR(I)**3
```

```
      IF (I.GE.4) GOTO20
      IF (ETAR(1).GT.0.0) GOTO19
      R(I)=SAVE
      GO TO 21
19    R(I)=SAVE-B3*ETAR(1)**3
      GO TO 21
20    R(I)=SAVE-B3*ETAR(4)**3
21    CONTINUE
      IF (I2.NE.0) GOTO28
C  FIN TIP MACH CONE CORRECTION
      LW=MU-S*CMU
      IF (LW.GE.C) GOTO26
      SL=C-LT
      IF (LW.GE.SL) GOTO24
      IF (LW.GE.LL) GOTO23
      PL(1)=(LL-LW)/LL
      PL(4)=PL(1)
      DO 22 I=2,3
      PL(I+3)=1.
22    PL(I)=1.
      GO TO 30
23    PL(2)=(C-LT-LW)/SM
      PL(5)=PL(2)
      PL(3)=1.
      PL(6)=PL(3)
      PL(1)=0.
      PL(4)=0.
      GO TO 30
24    PL(3)=(C-LW)/LT
      PL(6)=PL(3)
      DO 25 I=1,2
      PL(I)=0.
25    PL(I+3)=0.
      GO TO 30
26    DO 27 I=1,6
27    PL(I)=0.
      GO TO 30
28    DO 29 I=1,6
29    PL(I)=0.0
C  STRIP COEFFICIENTS
30    DO 31 I=1,3
31    FL(I)=-R(I)*D(I)*(1.-.5*PL(I))
      DO 32 I=4,6
32    FL(I)=R(I)*D(I)*(1.-.5*PL(I))
      DO 33 I=1,6
33    FD(I)=R(I)*N(I)*(1.-.5*PL(I))
      DO 34 I=1,4,3
34    XP(I)=XL
      DO 35 I=2,5,3
35    XP(I)=XL+D(I-1)
      DO 36 I=3,6,3
36    XP(I)=XL+D(I-1)+D(I-2)
      MMOMT=0.
      DO 37 I=1,6
      CW=.5*D(I)*(1.-PL(I)+.5*PL(I)**2+XP(I)*(2.-PL(I))/D(I))/(1.-.5*PL(
     1I))
37    MMOMT=MMOMT-CW*FL(I)
      LIFT=0.0
      DRAG=0.0
      DO 38 I=1,6
      LIFT=LIFT+FL(I)
38    DRAG=DRAG+FD(I)
      LMOM=S*LIFT
      TLIFT=TLIFT+LIFT
      TDRAG=TDRAG+DRAG
      TMMOM=TMMOM+MMOMT
```

```
      TLMOM=TLMOM+LMOM
      IF (S.LT.SPAN) GOTO17
C  TOTAL COEFFICIENTS
      CL=TLIFT*XNS
      CD=TDRAG*XNS
      CM=TMMOM*XNS/REFL
      CLM=2.*TLMOM*XNS/REFL
      CPX=TMMOM/(TLIFT*COS(ALPHAR))
      CPS=TLMOM/TLIFT
      IF (K.EQ.1) CDO(J)=CD
      IF (K.EQ.2) GOTO39
      GO TO 40
C LIFT CURVE SLOPE
39    CLA(J)=CL/DTR
      CLD(J)=CLM/DTR
      CMA(J)=CM/DTR
      CPXX(J)=CPX
      CPSS(J)=CPS
40    IF (AMAX.EQ.1.) GOTO42
      WRITE (6,41) ALPHA,CL,CM,CLM,CD,CPX,CPS
41    FORMAT (1H F9.3,4F8.4,2F9.3)
42    CONTINUE
      GO TO 11
43    J=J-1
      IF (AMAX.EQ.1.) GOTO45
      WRITE (6,44)
44    FORMAT (63H1    M      CLA/RAD.  CMA/RAD.  CLD/RAD.    CDW     CPX
     1      CPS)
      GO TO 47
45    WRITE (6,46)
46    FORMAT (63H0    M      CLA/RAD.  CMA/RAD.  CLD/RAD.    CDW     CPX
     1      CPS)
47    WRITE (6,48) (M(I),CLA(I),CMA(I),CLD(I),CDO(I),CPXX(I),CPSS(I),I=1
     1,J)
48    FORMAT (1H F7.3,F9.3,2F10.4,3F9.4)
      GO TO 1
      END
```

APPENDIX B

SAMPLE CASE

Figure B-1 shows the Tomahawk sounding rocket fin. For this run, the input data were chosen as follows.

M_o	=	3.0
ΔM	=	1.0
M_{max}	=	7.0
α_{max}	=	5.0 degrees
A	=	63.6 $in.^2$
L	=	9.0 in.
C_r	=	21.9 in.
S	=	13.8 in.
l_{Lr}	=	3.76 in.
l_{Tr}	=	0. in.
Γ_L	=	45 degrees
Γ_1	=	38.23 degrees
Γ_2	=	0. degrees
Γ_T	=	0. degrees
ζ_L	=	2.475 degrees
ζ_T	=	0. degrees
n	=	300
k	=	1

Figure B-2 shows a listing of the input cards.



Figure B-1. Tomahawk Fin

```
TOMAHAWK FIN INPUTS
3.       1.       7.       5.       63.6     9.       21.9     13.8
3.76     0.0      45.      38.23    0.0      0.0      2.475    0.0
300   1
```

Figure B-2. Tomahawk Fin Input Listing

The output resulting from these inputs is shown below.

```
TOMAHAWK FIN INPUTS

ZETA L =  2.475 DEG.     ZETA T =  0.     DEG.

AREF =     63.60  LREF =     9.000  STRIPS = 300
```

M = 3.000

ALPHA	CL	CM	CLM	CDW	CPX	CPS
0.	0.0000	-0.0000	0.0000	0.0029	-7.695	5.834
1.000	0.1655	-0.2519	0.2147	0.0058	-13.702	5.839
2.000	0.3313	-0.5040	0.4298	0.0146	-13.702	5.839
3.000	0.4976	-0.7566	0.6457	0.0293	-13.702	5.839
4.000	0.6648	-1.0098	0.8627	0.0499	-13.703	5.839
5.000	0.8332	-1.2638	1.0812	0.0767	-13.703	5.839

M = 4.000

ALPHA	CL	CM	CLM	CDW	CPX	CPS
0.	0.0000	-0.0000	0.0000	0.0021	-5.407	5.340
1.000	0.1219	-0.1847	0.1580	0.0043	-13.641	5.836
2.000	0.2442	-0.3699	0.3167	0.0108	-13.642	5.836
3.000	0.3674	-0.5562	0.4765	0.0217	-13.642	5.836
4.000	0.4921	-0.7442	0.6382	0.0372	-13.644	5.836
5.000	0.6187	-0.9345	0.8024	0.0572	-13.645	5.836

M = 5.000

ALPHA	CL	CM	CLM	CDW	CPX	CPS
0.	0.0000	-0.0000	0.0000	0.0017	-6.606	5.268
1.000	0.0972	-0.1466	0.1259	0.0035	-13.579	5.832
2.000	0.1950	-0.2940	0.2527	0.0087	-13.580	5.832
3.000	0.2941	-0.4431	0.3811	0.0175	-13.581	5.832
4.000	0.3951	-0.5949	0.5121	0.0300	-13.583	5.832
5.000	0.4987	-0.7500	0.6464	0.0463	-13.586	5.833

M = 6.000

ALPHA	CL	CM	CLM	CDW	CPX	CPS
0.	0.0000	-0.0000	0.0000	0.0015	-7.423	5.767
1.000	0.0812	-0.1219	0.1051	0.0029	-13.515	5.828
2.000	0.1631	-0.2449	0.2113	0.0073	-13.516	5.828
3.000	0.2467	-0.3701	0.3196	0.0148	-13.519	5.829
4.000	0.3328	-0.4988	0.4310	0.0254	-13.523	5.829
5.000	0.4221	-0.6320	0.5468	0.0394	-13.529	5.829

M = 7.000

ALPHA	CL	CM	CLM	CDW	CPX	CPS
0.	0.0000	-0.0000	0.0000	0.0013	-7.159	5.629
1.000	0.0699	-0.1045	0.0905	0.0026	-13.449	5.824
2.000	0.1408	-0.2104	0.1823	0.0064	-13.452	5.825
3.000	0.2137	-0.3191	0.2766	0.0129	-13.457	5.825
4.000	0.2894	-0.4319	0.3747	0.0222	-13.464	5.825
5.000	0.3691	-0.5505	0.4779	0.0346	-13.472	5.826

M	CLA/RAD.	CMA/RAD.	CLD/RAD.	CDW	CPX	CPS
3.000	9.482	-14.4327	12.3032	0.0029	-13.7018	5.8392
4.000	6.982	-10.5806	9.0540	0.0021	-13.6413	5.8356
5.000	5.567	-8.3980	7.2149	0.0017	-13.5786	5.8320
6.000	4.651	-6.9823	6.0232	0.0015	-13.5145	5.8282
7.000	4.007	-5.9866	5.1859	0.0013	-13.4494	5.8244

REFERENCES

1. Hilton, W. F.: High Speed Aerodynamics. Longmans, Green and Co. New York. 1951.

2. Pitts, W. C.; Nielsen, J. N.; and Kaattari, G. E.: Lift and Center of Pressure of Wind-Body-Tail Combinations at Subsonic, Transonic, and Supersonic Speeds. NACA Report 1307. 1959.

3. Shapiro, A. H.: The Dynamics and Thermodynamics of Compressible Fluid Flow. Vol. II. Ronald Press. New York. 1954.

4. Harmon, S. M.; and Jeffreys, I.: Theoretical Lift and Damping in Roll of Thin Wings with Arbitrary Sweep and Taper at Supersonic Speeds Supersonic Leading and Trailing Edges. NACA TN 2114. May 1950.

5. Thomas, E. S.: Windtunnel Tests of the Aerobee 150A. Aerojet General Corp. 1960.

6. Mc Nerney, J. D.: Aerobee 350 Windtunnel Test Analysis. Space General Corp. January 1963.